
 Introductions
 Name

 How much APL experience? 

 Goals

 Agenda
60 15 60 15 60

 Participate
 Ask questions

 Work together

 Ask for help if you need it

 Have fun

 (⎕IO ⎕ML)←1

Welcome!

≢Y

Tally

avg ← +/÷⍴  ⍝ {(+/⍵)÷⍴⍵}
avg 3 1 4 1 5 9 2 6

3.875
avg 42

avg 3 4⍴⍳12
LENGTH ERROR

Just an average APL expression

avg ← +⌿÷≢ ⍝ {(+⌿⍵)÷≢⍵}
avg 3 1 4 1 5 9 2 6

3.875
avg 42

42
avg 3 4⍴⍳12

5 6 7 8

Just an average APL expression

≢'Mississippi'
11

⍴'Mississippi'
11

]display ≢'Mississippi'
11

]display ⍴'Mississippi'
┌→─┐
│11│
└~─┘

Tally versus Shape

⊢scalar←42
42

⍴scalar

≢scalar
1

Tally versus Shape

Vector of 

Dimensions

Scalar

Count

⊢mat←3 4⍴⍳12
1  2  3  4
5  6  7  8  
9 10 11 12
 ⍴mat
3 4

      ≢mat
3

Tally versus Shape

Vector of 

Dimensions

Scalar

Count



Selected Primitives8

≢2 3⍴⎕A
2

⍴2 3⍴⎕A
2 3

≢4 5 6 7 8 ⍴ ⍳6720
4

Tally: Leading Axis

1. Given a matrix and an array which may be either a vector or 
a matrix, write a function to conditionally concatenate the 
array to the matrix if their shapes are conforming.

      
      

2. Write an expression to test if an array A is empty.

Tally: Exercises

(2 3⍴⍳6){fn}'ab'
1 2 3 a
4 5 6 b

(2 3⍴⍳6){fn}'abc'
1 2 3
4 5 6

Note: The solutions presented in this workshop are not the only possible solutions to the 
exercises

1. {⍺=⍥≢⍵: ⍺,⍵ ⋄ ⍺}
{(≢⍺)=≢⍵: ⍺,⍵ ⋄ ⍺}

2. 0∊⍴A
Why isn't ≢ used? Because it may be that a non-leading dimension is 0.
      A←3 0 4⍴42
      0=≢A
0
      0∊⍴A
1

Tally: Exercises Solutions

≠Y

Unique Mask

∪'Mississippi'
Misp
      ≠'Mississippi'
1 1 1 0 0 0 0 0 1 0 0

      ↑'Mississippi' (≠'Mississippi')
M i s s i s s i p p i
1 1 1 0 0 0 0 0 1 0 0

      (≠'Mississippi') ⌿ 'Mississippi'
Misp

Unique Mask

(≠'Mississippi') ⌿ 'Mississippi'
Misp      
      (≠⌿⊢)'Mississippi'
1

We'll come back to this soon

Unique Mask

≠m

1 1 0 0

      (≠m) ⌿ m

 1  2  3
 4  5  6

 7  8  9
10 11 12

Unique Mask – Leading Axis
⊢m←(4 2 3⍴⍳12)

 1  2  3
 4  5  6

 7  8  9
10 11 12

 1  2  3
 4  5  6

 7  8  9
10 11 12

≠ 'Mississippi'
1 1 1 0 0 0 0 0 1 0 0

      ~≠ 'Mississippi'
0 0 0 1 1 1 1 1 0 1 1

      (~≠'Mississippi') ⌿ 'Mississippi'
sissipi

Duplicate Elements

mat ← ? 1000 1000 ⍴ 1000000

      ]runtime -c +/,mat +/∊mat +/+/mat +⌿+⌿mat

  +/,mat  → 5.3E¯4 |   0% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷
  +/∊mat  → 5.2E¯4 |  -1% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷
  +/+/mat → 2.0E¯4 | -64% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷
  +⌿+⌿mat → 6.5E¯4 | +22% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷

Tangent - ]runtime

d← ? 1e6 ⍴ 1e6

      (≢∪d) (+/≠d)
632453 632453

      ]runtime -c ≢∪d +/≠d
     
  ≢∪d  → 9.7E¯3 |  0% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷    
  +/≠d → 1.0E¯2 | +7% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷ 

Count Uniques

1. Oh no!  Your ∪ key is broken, and there's no language bar 
available, and you can't copy and paste… Write a function 
to return the unique duplicated letters in a word.

      {your-function}'Mississippi'

sip

Unique Mask Exercises

1. (≠⊢⍤/⊢)(~∘≠⊢⍤/⊢)
{{⍵/⍨≠⍵}⍵/⍨~≠⍵}

Unique Mask Exercise Solutions

X⊣Y
X⊢Y
 ⊣Y
 ⊢Y

Left/Right/Same

23 ⊣ 42 ⍝ left
23
      23 ⊢ 42 ⍝ right
42
         ⊣ 42 ⍝ same
42 
         ⊢ 42 ⍝ same
42

Left/Right/Same

⊣ ⊢

Monadic {⍵} {⍵}

Dyadic {⍺} {⍵}

Ambivalent {⍺←⍵ ⋄ ⍺}
{⍵}⍨
⊢⍨

{⍵}
 {⍺}⍨
⊣⍨

Left/Right/Same

    (rc msg)←¯1 'oops' ⋄ →0

    →0 ⊣ (rc msg)←¯1 'oops'

The "Anti-diamond"

'myapp.log' Log 'Test failed!'
LOGGED: Test failed!

      Log←{⍵ ⎕NPUT ⍺ 2 ⋄ LOGGED: ',⍵}

      Log←{'LOGGED: ',⍵ ⊣ ⍵ ⎕NPUT ⍺ 2}

The "Anti-diamond"

Append←{ tn←⍺ ⎕NTIE 0
               (⎕NUNTIE tn) ⊢ ⍵ ⎕NAPPEND tn }

      '/tmp/foo.msgs' Append 'my message'
10

Replace one result with another

'[ace]' ⎕R 'x' 'abcde'
┌─────┬──┬─────────┐
│[ace]│⎕R│┌─┬─────┐│
│     │  ││x│abcde││
│     │  │└─┴─────┘│
└─────┴──┴─────────┘
      '[ace]' ⎕R 'x'⊢'abcde'
xbxdx

Separate right operand and argument

'[ace]' ⎕R 'x'⊢'abcde'
xbxdx

      ('[ace]' ⎕R 'x')'abcde'
xbxdx

Separate right operand and argument

⊢   refers to the right argument
      (⊂∘⍋⌷⊢) 'sort me'
 emorst

⊣   refers to the left argument 
      'split-me-on' (≠⊆⊣) '-'
┌─────┬──┬──┐
│split│me│on│
└─────┴──┴──┘

Address arguments in tacit functions

'12,3,456' {(⍺≠⍵)⊆⍺} ','
┌──┬─┬───┐
│12│3│456│
└──┴─┴───┘ 

Address arguments in tacit functions

{(⍺≠⍵)⊆⍺} ⍝ replace {} with ()

Address arguments in tacit functions

((⍺≠⍵)⊆⍺) ⍝ (⍺ f ⍵) → f
          

Address arguments in tacit functions

(≠⊆⍺) ⍝ ⍺ → ⊣
          

Address arguments in tacit functions

(≠⊆⊣)
          

Address arguments in tacit functions

      '12,3,456'(≠⊆⊣)','
┌──┬─┬───┐
│12│3│456│
└──┴─┴───┘

          

Address arguments in tacit functions

(⊢-⌊⌿)N Normalise so minimum is 0

(∪⍳⊢)Y Consecutive IDs

(⊢≡∪)Y
 All distinct 
(∪≡⊢)Y 

Hooks: X f (g X) and (f X) g X

⎕VFI '23 daffy 42'
┌─────┬───────┐
│1 0 1│23 0 42│
└─────┴───────┘
      // ⎕VFI '23 daffy 42'
┌─────┐
│23 42│
└─────┘
      ⊃// ⎕VFI '23 daffy 42'
LENGTH ERROR

Force hybrid (/⌿\⍀) to be a function

      // ⎕VFI '23 daffy 42'
┌─────┐
│23 42│
└─────┘
      ⊃// ⎕VFI '23 daffy 42'
LENGTH ERROR

Force hybrid (/⌿\⍀) to be a function

Replicate Reduce

⊃// ⎕VFI '23 daffy 42'
LENGTH ERROR

      ⊃⊢⍤// ⎕VFI'23 daffy 42'
23 42

Force hybrid (/⌿\⍀) to be a function

(≠'Mississippi') ⌿ 'Mississippi'
Misp      
      (≠⌿⊢)'Mississippi'
1
      (≠⊢⍤⌿⊢)'Mississippi'
Misp

Remember Unique Mask?

2 ⎕NQ # 'GetEnvironment' 'MAXWS'

      ⊢2 ⎕NQ # 'GetEnvironment' 'MAXWS'
4G
      vec ← ⍳10
      ⊢vec ← ⍳10
1 2 3 4 5 6 7 8 9 10

Making values non-shy

Env←2 ⎕NQ # 'GetEnvironment',⊂
      Env 'MAXWS'

      Env←⊢2 ⎕NQ # 'GetEnvironment',⊂
      Env 'MAXWS'
4G

Making values non-shy

Env←{2 ⎕NQ # 'GetEnvironment' ⍵}
      Env 'MAXWS'

      Env←{⊢2 ⎕NQ # 'GetEnvironment' ⍵}
      Env 'MAXWS'
4G

Making values non-shy



Selected Primitives43

⊢vec←⎕A
ABCDEFGHIJKLMNOPQRSTUVQXYZ

      ⊣/vec
A
      ⊢/vec
Z

First / Last

      ⊣⌿mat
abc
      ⊢⌿mat
ghi

      ⊣/mat
adg
      ⊢/mat
cfi

First / Last of the First / Last Axis

mat
abc
def
ghi

mat
abc
def
ghi

Using ]runtime and mat←?100 100⍴1E6
 Compare at least 3 ways to return the last column of mat as a 

vector

 Make sure you use ⊢/mat as one of your choices
      ]runtime -c [your solutions here]

  ⊢/mat        → 1.4E¯7 |     0% ⌷
  ,¯1↑[2]mat   → 8.6E¯7 |  +523% ⌷⌷⌷⌷⌷⌷⌷⌷
  mat[;2⊃⍴mat] → 3.5E¯7 |  +154% ⌷⌷⌷
  (⌽mat)[;1]   → 4.2E¯6 | +2973% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷

Exercise - Last column

⊢/mat ,¯1↑[2]mat mat[;2⊃⍴mat] (⌽mat)[;1]

Div←{0=⎕NC'⍺': ÷⍵ ⋄ ⍺÷⍵}
      Div←{⍺←⊢ ⋄ ⍺÷⍵}

      Div 3
0.3333333333

      4 Div 3
1.333333333

Propagate missing left argument of dfn

Modify a global variable:
      g←10
      ∇ foo y
        g←2×y
      ∇
      foo 3
      g
6

      g←10
      {g←2×⍵} 3
      g
10
      {g⊢←2×⍵} 3
      g
6

Counteracting dfn auto-localization

Use a pass-through value:
      F←-
      ∇ r←Use y
        r←y,F y←2×y
      ∇
      Use 3
6 ¯6

      F←-
      {y,F y←2×⍵} 3
6 6
      {y,F⊢y←2×⍵} 3
6 ¯6

Counteracting dfn auto-localization

Modified assignment:
      g←10 ⋄ F←-
      ∇ Mod y
        g F←y
      ∇
      Mod 2
      g
8

      g←10 ⋄ F←-
      {g F←⍵} 2
      g
10
      {g F∘⊢←⍵} 2
      g
8

Counteracting dfn auto-localization

'ABC'∘.⊢'abcd'
abcd
abcd
abcd
      'ABC'∘.⊣'abcd'
AAAA
BBBB
CCCC

Create matrix of identical rows/columns

true_case ⊣⍣(condition)⊢ false_case

      'true'⊣⍣(3>4)⊢'false'
false
      'true'⊣⍣(9>4)⊢'false'
true

Inline "if"

Given 2 matrices, M and N, where M has the same number of 
rows as N has columns – write an expression to multiply the 
last column of M by the first row of N.
      1  2  3  (your-expression) 1 2 3 4
      4  5  6                    5 6 7 8
      7  8  9
     10 11 12

3 12 27 48

Left/Right/Same Exercise

1. (⊢/⍤⊣×⊣⌿⍤⊢)
{(⊢/⍺)×⊣⌿⍵}

Left/Right/Same Exercise Solution

⍸Y

Where

Y

⍸Y

⍳⍴Y

(,Y)/(,⍳⍴Y)

1 0 1 1 0 0 1 0 1

1 3 4 7 9

1 2 3 4 5 6 7 8 9

1   3 4     7   9

Where: definition

Y

⍸Y

⍳⍴Y

(,Y)/(,⍳⍴Y)

1 0 1 1 0 0 1 0 1

1 3 4 7 9

1 2 3 4 5 6 7 8 9

1 3 4 7 9

Where: definition

0≤Y

  m
1 0 1
1 1 0

      

     ⍳⍴m
┌───┬───┬───┐
│1 1│1 2│1 3│
├───┼───┼───┤
│2 1│2 2│2 3│
└───┴───┴───┘

Where: Rank > 1

m ← 2 3 ⍴ 1 0 1 1 1 0

,m         1   0   1   1   1   0 
          ┌───┬───┬───┬───┬───┬───┐
     ,⍳⍴m │1 1│1 2│1 3│2 1│2 2│2 3│
          └───┴───┴───┴───┴───┴───┘
          ┌───┬───┬───┬───┐
(,m)/,⍳⍴m │1 1│1 3│2 1│2 2│
          └───┴───┴───┴───┘

          ┌───┬───┬───┬───┐
       ⍸m │1 1│1 3│2 1│2 2│
          └───┴───┴───┴───┘

Where: Rank > 1

⍸ 1 0 2 0 3
1 3 3 5 5 5
      ⊢ m ← 2 3 ⍴ 0 2 1 3 0 1
0 2 1
3 0 1
      ⍸m
┌───┬───┬───┬───┬───┬───┬───┐
│1 2│1 2│1 3│2 1│2 1│2 1│2 3│
└───┴───┴───┴───┴───┴───┴───┘

Where: Y > 1

⍸5  ⍝ what will this produce?
┌┬┬┬┬┐
││││││
└┴┴┴┴┘
      ⍸,5 ⍝ how about this?
1 1 1 1 1

Where: what about?

price←71 82 81 82 84 59
      (75≤price)/price
82 81 82 84
      (75≤price)/⍳⍴price
2 3 4 5
      ⍸75≤price
2 3 4 5
      price[⍸75≤price]
82 81 82 84

Where: usage

fruit←'Apple' 'Banana' 'Cherry' 'Date'
      select←1 1 0 1

      select/fruit
 Apple  Banana  Date 
      ⍸select
1 2 4
      fruit[⍸select]
 Apple  Banana  Date

Where: Selection

fruit←'Apple' 'Banana' 'Cherry' 'Date'
      select←1 2 0 1

      select/fruit
 Apple  Banana  Banana  Date 

      fruit[⍸select]
 Apple  Banana  Banana  Date 

Where: Multi-selection

spice ← 'Anise' 'Basil' 'Chili' 'Dill'
      ⊢stuff← ↑fruit spice
 Apple  Banana  Cherry  Date 
 Anise  Basil   Chili   Dill
      ⊢select←↑(1 2 0 1) (0 0 0 2)
1 2 0 1 0
0 0 0 2 0
      select/stuff
RANK ERROR

Where: Multi-dimensional selection

spice ← 'Anise' 'Basil' 'Chili' 'Dill' 
⊢stuff← ↑fruit spice

 Apple  Banana  Cherry  Date
 Anise  Basil   Chili   Dill
      ⊢select←↑(1 2 0 1) (0 0 0 2)
1 2 0 1 0
0 0 0 2 0
      stuff[⍸select]
 Apple  Banana  Banana  Date  Dill  Dill

Where: Multi-dimensional selection

⍸ Y       ⍝ counts  → indices
      ⍸ 2 0 2
1 1 3 3

      ⍸⍣¯1⊢ Y   ⍝ indices → counts
      ⍸⍣¯1⊢ 1 1 3 3
2 0 2

Where Inverse: ⍸⍣¯1

all     ← 'a' 'b' 'c' 'd' 'e' 'f'
      count   ←  1   0   0   3   0   2
      indices ←  1         4 4 4    6 6
      indices ≡ ⍸ count
1
      count   ≡ ⍸⍣¯1⊢indices
1

Where: Representing a multiset

all     ← 'a' 'b' 'c' 'd' 'e' 'f' 'g'
      count   ←  1   0   0   3   0   2   0
      indices ←  1         4 4 4    6 6
      indices ≡ ⍸ count
1
      count   ≡ ⍸⍣¯1⊢indices
0
      ⍸⍣¯1⊢indices
1 0 0 3 0 2
      count   ≡ (⍴count) ↑ ⍸⍣¯1⊢indices
1

Where Inverse:

?

data ← 'banana'
      (∪∘.≡⊢) data
1 0 0 0 0 0
0 1 0 1 0 1
0 0 1 0 1 0

Where Inverse: Reconstruction of masks

data ← 'banana'
      ('≡',∪data) , data ⍪ (∪∘.≡⊢)data
≡ b a n a n a
b 1 0 0 0 0 0
a 0 1 0 1 0 1 
n 0 0 1 0 1 0

Where Inverse: Reconstruction of masks

⊢⌸ d←'banana'
1 0 0
2 4 6
3 5 0
      ⍸⍣¯1⍤⊢⌸ d
1 0 0 0 0 0
0 1 0 1 0 1
0 0 1 0 1 0

{⍸⍣¯1⊢⎕←⍵}⍤⊢⌸ d
1
2 4 6
3 5
1 0 0 0 0 0
0 1 0 1 0 1
0 0 1 0 1 0

Where Inverse: Reconstruction of masks

data ← 10000⍴'Mississippi'
      ]runtime –c (∪∘.≡⊢)data  ⍸⍣¯1⍤⊢⌸data
                                                                       
   (∪∘.≡⊢)data → 2.5E¯3 |   0% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷
   ⍸⍣¯1⍤⊢⌸data → 2.4E¯4 | -91% ⌷⌷

Where Inverse: Reconstruction of masks

Given a Boolean vector (v←¯1+?100⍴2), write an expression 
using ⍸ (twice!) to return the highest number of consecutive 
1's.

      (yours) 1 0 1 1 1 0 1 1 1 1 0 1 1 0 1
4

      ⌈/¯2-/⍸1≠0,⍨¯2-/⍸      

Where: Exercise

V←1 0 1 1 1 0 1 1 1 1 0 1 1 0 1

      ⌈/¯2-/⍸1≠0,⍨¯2-/⍸V
4

Where: Exercise Solution

X⍸Y

Interval Index

┌────────┬─┬─┐
       ┌───│────┬─┐ │ │ │
     ┌─│───│──┐ │ │ │ │ │
      2 4 5 ⍸ 1 2 3 7 8 9     ⁰ ¹ ² ³
┌────┘ │   │
│ ┌─┬──┘   │
│ │ │ ┌─┬─┬┘
0 1 1 3 3 3

Interval Index

Breaks       ⍸  Data
        10 20 30      ⍸  23 2 42 7 5 19 24 
¯∞ ←--10---20---30--→ ∞  Breaks define buckets
    0    1    2    3     Buckets/Intervals
                         (⎕IO←1)

Interval Index – the bucket list primitive

0 1 2 3
[¯∞,10) [10,20) [20,30) [30,∞]
D<10 D<.<10 20 D<.<20 30 D≥30

2 4 5 ⍸ 1 2 3 5 7 8 9
0 1 1 3 3 3 3
      2 4 5 ∊⍨ 1 2 3 5 7 8 9
0 1 0 1 0 0 0
      2 4 5 (⍸-∊⍨) 1 2 3 5 7 8 9
0 0 1 2 3 3 3

What if you want (lower,upper]?

2 5 ∘.≤ 1 2 3 4 5 6
0 1 1 1 1 1
0 0 0 0 1 1
      >⌿2 5 ∘.≤ 1 2 3 4 5 6
0 1 1 1 0 0
      1=2 5 ⍸ 1 2 3 4 5 6 
0 1 1 1 0 0
      2 5 (1=⍸) 1 2 3 4 5 6
0 1 1 1 0 0

Check if value is in interval

┌────────┬─┬─┐
           ┌───│────┬─┐ │ │ │
     ┌─────│───│──┐ │ │ │ │ │
      2 2 2 4 5 ⍸ 1 2 3 7 8 9     ⁰ ¹ ² ³ ⁴ ⁵
┌────┘     │   │
│ ┌─┬──────┘   │
│ │ │ ┌─┬─┬────┘
0 3 3 5 5 5
      2 2 2 4 5 ∘.≤ 1 2 3 7 8 9
0 1 1 1 1 1
0 1 1 1 1 1
0 1 1 1 1 1
0 0 0 1 1 1
0 0 0 1 1 1

Non-unique cut-offs

┌────────┬─┬─┐
           ┌───│────┬─┐ │ │ │
     ┌─────│───│──┐ │ │ │ │ │
      2 2 2 4 5 ⍸ 1 2 3 7 8 9     ⁰ ¹ ² ³ ⁴ ⁵
┌────┘     │   │
│ ┌─┬──────┘   │
│ │ │ ┌─┬─┬────┘
0 3 3 5 5 5
   +⌿ 2 2 2 4 5 ∘.≤ 1 2 3 7 8 9
0 3 3 5 5 5

Non-unique cut-offs



Selected Primitives82

n←?⍨10000 
      c←{⍵[⍋⍵]}100?10000

      ]runtime –c +⌿c∘.≤n c⍸n

  +⌿c∘.≤n → 9.9E¯5 |   0% ⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷⌷ 
  c⍸n     → 9.2E¯6 | -91% ⌷⌷⌷⌷

Performance

qtrs
2024  1 15
2024  4 15
2024  7 15
2024 10 15

qtrs⍸dts
1 3 3 4 3 2

dts
2024  3  6
2024  9  8
2024  7 28
2024 10 21
2024  8 15
2024  7 11

Buckets of higher rank

⊢qtrs←↓qtrs
┌─────────┬─────────┬─────────┬──────────┐
│2024 1 15│2024 4 15│2024 7 15│2024 10 15│
└─────────┴─────────┴─────────┴──────────┘
      ⊢dts←↓dts
┌────────┬────────┬─────────┬──────────┬───────
│2024 3 6│2024 9 8│2024 7 28│2024 10 21│202 ...
└────────┴────────┴─────────┴──────────┴───────
      qtrs ⍸ dts
1 3 3 4 3 2 4 4 1 1

Total Array Ordering

'Abe' 'Bob' 'Carla' 'Dana'⍸'Bea' 'Aaron'
1 0

Total Array Ordering

⊢table←'Abe' 'Abe' 'Abe' 'Bob',⍪1 3 9 2
┌───┬─┐
│Abe│1│
├───┼─┤
│Abe│3│
├───┼─┤
│Abe│9│
├───┼─┤
│Bob│2│
└───┴─┘

Total Array Ordering

new← 'Abe' 4
      pos← table⍸new
2
      pos↑table
┌───┬─┐
│Abe│1│
├───┼─┤
│Abe│3│
└───┴─┘

      pos↓table
┌───┬─┐
│Abe│9│
├───┼─┤
│Bob│2│
└───┴─┘

Total Array Ordering
table

┌───┬─┐
│Abe│1│
├───┼─┤
│Abe│3│
├───┼─┤
│Abe│9│
├───┼─┤
│Bob│2│
└───┴─┘

pos(↑⍪new⍪↓)table
┌───┬─┐
│Abe│1│
├───┼─┤
│Abe│3│
├───┼─┤
│Abe│4│
├───┼─┤
│Abe│9│
├───┼─┤
│Bob│2│
└───┴─┘

Total Array Ordering

'AL'⍸'Dan' 'Sam' 'Ann' 'Tim' 'Bea'
1 2 1 2 1
      n←'Dan' 'Sam' 'Ann' 'Tim' 'Bea'
      ('AL'⍸n){⊂⍵}⌸n
┌─────────────┬─────────┐
│┌───┬───┬───┐│┌───┬───┐│
││Dan│Ann│Bea│││Sam│Tim││
│└───┴───┴───┘│└───┴───┘│
└─────────────┴─────────┘

Group by interval

Many schools convert numeric scores into a letter grade. Scotland uses:

Interval Index Exercise

Percent Grade Description
70–100% A Highest grade
60–69% B Very strong pass
50–59% C Pass and be accepted by universities
40–49% D Borderline: the student should re-sit the course 
0–39% No Award The student has failed the course

Write a function using ⍸ that will return the grade for a given percentage.

This solution also deals with scores that are less than 0 or greater than 100.

 GradeFor←{
     grades←,¨'Invalid' 'No Award' 'D' 'C' 'B' 'A' 'Invalid'
     grades⊃⍨1+(100=⍵)-⍨0 40 50 60 70 100⍸⍵
 }

       GradeFor ¨¯1 0 39 40 49 50 59 60 69 70 99 100 101
Invalid  No Award  No Award  D  D  C  C  B  B  A  A  A  Invalid

Interval Index Exercise Solution

f⌸

Key

Monadic Operator

Analogous to SQL "GROUP BY"

SELECT (dates,amounts) FROM sales GROUP BY customer

Key

{⍺ ⍵}⌸'Abracadabra'
┌─┬────────┐
│A│1       │
├─┼────────┤
│b│2 9     │
├─┼────────┤
│r│3 10    │
├─┼────────┤
│a│4 6 8 11│
├─┼────────┤
│c│5       │
├─┼────────┤
│d│7       │
└─┴────────┘
 │ └─────────────── ⍵: indices where ⍺ occurs
 │
 └──────────────────── ⍺: ∪Y

Monadic Key: f⌸Y

'Abracadabra'{⍺ ⍵}⌸'ABCDEFGHIJK'
┌─┬────┐
│A│A   │
├─┼────┤
│b│BI  │
├─┼────┤
│r│CJ  │
├─┼────┤
│a│DFHK│
├─┼────┤
│c│E   │
├─┼────┤
│d│G   │
└─┴────┘
 │ └───────────────── ⍵: elements of Y where ⍺ occurs
 │
 └──────────────────── ⍺: ∪X

Dyadic Key:  Xf⌸Y

'Abracadabra'{⍺ ⍵}⌸⍳≢'Abracadabra'
┌─┬────────┐
│A│1       │
├─┼────────┤
│b│2 9     │
├─┼────────┤
│r│3 10    │
├─┼────────┤
│a│4 6 8 11│
├─┼────────┤
│c│5       │
├─┼────────┤
│d│7       │
└─┴────────┘

Dyadic Key

{⍺ ⍵}⌸  'Abracadabra'

(⎕C'Abracadabra'){⍺ ⍵}⌸'Abracadabra'
┌─┬─────┐
│a│Aaaaa│
├─┼─────┤
│b│bb   │
├─┼─────┤
│r│rr   │
├─┼─────┤
│c│c    │
├─┼─────┤
│d│d    │
└─┴─────┘

Key: grouping by an aspect of the data

⊢v←?10⍴20
2 11 19 14 1 14 3 8 1 16

      (3|v){⍺ ⍵}⌸v
┌─┬────────────┐
│2│2 11 14 14 8│
├─┼────────────┤
│1│19 1 1 16   │
├─┼────────────┤
│0│3           │
└─┴────────────┘

Key: grouping by an aspect of the data

v←?1000⍴10
      {⍺ ('⎕'⍴⍨⌊.5+.1×≢⍵)}⌸v
 4  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕  
10  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕  
 2  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕  
 8  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕  
 1  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕ 
 5  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕ 
 9  ⎕⎕⎕⎕⎕⎕⎕⎕⎕   
 3  ⎕⎕⎕⎕⎕⎕⎕⎕⎕   
 6  ⎕⎕⎕⎕⎕⎕⎕⎕⎕   
 7  ⎕⎕⎕⎕⎕⎕⎕⎕⎕⎕ 

f⌸ - What the f? 

t←'Herebedragons'
      p←1 1 1 1 2 2 3 3 3 3 3 3 3

      p{⊂⍵}⌸t
┌────┬──┬───────┐
│Here│be│dragons│
└────┴──┴───────┘
      p⊆t
┌────┬──┬───────┐
│Here│be│dragons│
└────┴──┴───────┘

Key versus Partition

t←'Herebedragons'
      p←3 3 3 3 2 2 3 3 3 3 3 3 3

      p{⊂⍵}⌸t
┌───────────┬──┐
│Heredragons│be│
└───────────┴──┘
      p⊆t
┌──────┬───────┐
│Herebe│dragons│
└──────┴───────┘

Key versus Partition

{⍺⍵}⌸'abracadabra'
┌─┬──────────┐
│a│1 4 6 8 11│
├─┼──────────┤
│b│2 9       │
├─┼──────────┤
│r│3 10      │
├─┼──────────┤
│c│5         │
├─┼──────────┤
│d│7         │
└─┴──────────┘          

Key versus Unique
{⍺}⌸'abracadabra'

abrcd

   ∪'abracadabra'
abrcd

d←'2718281828'

    {⍺,≢⍵}⌸d
2 3
7 1
1 2
8 4

{⍺,≢⍵}⌸⎕D,d
0 1
1 3
2 4
3 1
4 1
5 1
6 1
7 2
8 5
9 1

Key issue: missing keys

d←'2718281828'

    {⍺,≢⍵}⌸d
2 3
7 1
1 2
8 4

{⍺,¯1+≢⍵}⌸⎕D,d
0 0
1 2
2 3
3 0
4 0
5 0
6 0
7 1
8 4
9 0

Key issue: missing keys

v←?10⍴20
      (3|v){⍺ ⍵}⌸⍳≢v
┌─┬──────────┐
│1│1 6 7 8 10│
├─┼──────────┤
│2│2 4 9     │
├─┼──────────┤
│0│3 5       │
└─┴──────────┘

sort←⊂⍤⍋⌷⊢
      sort(3|v){⍺ ⍵}⌸⍳≢v
┌─┬──────────┐
│0│3 5       │
├─┼──────────┤
│1│1 6 7 8 10│
├─┼──────────┤
│2│2 4 9     │
└─┴──────────┘

Key issue: sorting keys

(3|v){⍺ ⍵}⌸⍳≢v
┌─┬──────────┐
│1│1 6 7 8 10│
├─┼──────────┤
│2│2 4 9     │
├─┼──────────┤
│0│3 5       │
└─┴──────────┘

Key issue: sorting keys

(0 1 2,3|v){⍺ ⍵}⌸0 1 2,⍳≢v
┌─┬────────────┐
│0│0 3 5       │
├─┼────────────┤
│1│1 1 6 7 8 10│
├─┼────────────┤
│2│2 2 4 9     │
└─┴────────────┘

Key issue: sorting keys

(0 1 2,3|v){⍺ ⍵}⌸0 1 2,⍳≢v
┌─┬────────────┐
│0│0 3 5       │
├─┼────────────┤
│1│1 1 6 7 8 10│
├─┼────────────┤
│2│2 2 4 9     │
└─┴────────────┘

Key issue: sorting keys

(0 1 2,3|v){⍺ (1↓⍵)}⌸0 1 2,⍳≢v
┌─┬──────────┐
│0│3 5       │
├─┼──────────┤
│1│1 6 7 8 10│
├─┼──────────┤
│2│2 4 9     │
└─┴──────────┘

Key issue: sorting keys

You may have heard there's an election coming up in the USA. 
Most polling locations have designated tables for ranges of the 
alphabet. 

For instance, if surname/last name happens to be "Becker", 
you need to go to table 1 to have your registration validated. 

Key Exercise

Write a function to group a list of non-case sensitive surnames 
by table as follows:

Key Exercise

Surname Begins With Go To Table
A–H & a–h 1
I–O & i–o 2
P–Z & p–z 3

Anything else 4

This solution returns a matrix sorted by table number with each row containing the list of names assigned to each table.

      tables←{((⍳4),4@(=∘0)⊢{('z'=⍵)-⍨'aipz'⍸⍵}⎕C⊃¨⍵){⍺ (1↓⍵)}⌸1 2 3 4,⊆⍵}
      tables '123' 'becker' 'smith' 'zither' 'ørson'
1   becker         
2                  
3   smith  zither  
4   123  ørson 

This solution returns a matrix not sorted by table number and only rows where a name exists for a particular table.

      tables2←{(4@(=∘0)⊢,{('z'=⍵)-⍨'aipz'⍸⍵}⎕C⊃¨⍵){⍺⍵}⌸,⍵} 
      tables2'123' 'becker' 'smith' 'zither' 'ørson'

4   123  ørson     
1   becker         
3   smith  zither 

Key Exercise Solution

tables←{((⍳4),4@(=∘0)⊢{('z'=⍵)-⍨'aipz'⍸⍵}⎕C⊃¨⍵){⍺ (1↓⍵)}⌸1 2 3 4,⊆⍵}

      tables '123' 'becker' 'smith' 'zither' 'ørson'
1   becker         
2                  
3   smith  zither  
4   123  ørson 

      tables2←{(4@(=∘0)⊢,{('z'=⍵)-⍨'aipz'⍸⍵}⎕C⊃¨⍵){⍺⍵}⌸,⍵} 
      tables2'123' 'becker' 'smith' 'zither' 'ørson'

4   123  ørson     
1   becker         
3   smith  zither 

Key Exercise Solution

@

At

{X}(what @ where)Y

2(*⍨ @ 2 4 6)⍳6

1 4 3 16 5 36

At

'*' @ 1 3 5 'Hello'
┌─┬─┬─────────────┐
│*│@│┌─┬─┬─┬─────┐│
│ │ ││1│3│5│Hello││
│ │ │└─┴─┴─┴─────┘│
└─┴─┴─────────────┘
      '*' @ 1 3 5 ⊢ 'Hello'
*e*l*
      ('*' @ 1 3 5) 'Hello'
*e*l*

At: Separating the argument

str←'Hello'             
      mask←str∊'lo'
      (mask/str)←'*'
      'abc ',str
abc He***

      'abc ','*'@(∊∘'lo')⊢'Hello'
abc He***

At: versus Selective Assignment

'abc ', 'lo'{(⍵∊⍺)/⍵)←'*'}'Hello'
SYNTAX ERROR

      'abc ', 'lo'{w←⍵ ⋄ ((w∊⍺)/w)←'*'}'Hello'
abc *

At within dfns

'abc ', 'lo'{w←⍵⋄((w∊⍺)/w)←'*'⋄w}'Hello'
abc He***

      'abc ','lo' {'*'@(∊∘⍺)⊢⍵} 'Hello'
abc He***

At within dfns

23,(⍳10)[3 5 7] ← 42
SYNTAX ERROR
      vec←⍳10
      23,vec[3 5 7]←42
23 42
      23,vec
23 1 2 42 4 42 6 42 8 9 10
      23,42@3 5 7 ⊢ ⍳10
23 1 2 42 4 42 6 42 8 9 10

At versus Indexed Assignment

vec←5⍴0
      ⊢inds←?20⍴5
5 3 3 3 2 4 1 4 1 2 2 3 2 5 3 5 1 1 4 1
      vec[inds]+←1
      vec
5 4 5 3 3
      1+@inds⊢5⍴0
1 1 1 1 1

At versus Modified Assignment

(array @ array)

(array @ function)

(function @ array)

(function @ function)

The whats and wheres of @

r1←10×⍳3
      42 @ 2 ⊢ r1
10 42 30

      r2←2 3⍴10×⍳6
      42 @ 2 ⊢ r2
10 20 30
42 42 42

r3←2 3 3⍴10×⍳18
      42 @ 2 ⊢ r3
10 20 30
40 50 60

42 42 42
42 42 42

Major Cell Orientation

f is applied to the sub-array defined by g
      +/(⍳10)[2 3 4 6 9]
24
      +/@2 3 4 6 9 ⊢ ⍳10
1 24 24 24 5 24 7 8 24 10

When the left operand (f) is a function…

+\(⍳10)[2 3 4 6 9]
2 5 9 15 24

When the left operand (f) is a function…

+\(⍳10)[2 3 4 6 9]
  2 5 9   15     24
      +\@2 3 4 6 9 ⊢ ⍳10
1 2 5 9 5 15 7 8 24 10

When the left operand (f) is a function…

N is a scalar or vector of indices into Y
  
f is a function to be applied on the sub-array defined by N

      42 (×@ 3 5 7) ⍳10
1 2 126 4 210 6 294 8 9 10
      42×(@3 5 7) ⍳10
      42×@3 5 7 ⊢ ⍳10

{X}(f @ N) Y

⊢m←2 4⍴4/2 3
2 2 2 2
3 3 3 3      
      m×@2 4 ⊢ d

⊢d←4 4⍴⍳16
 1  2  3  4
 5  6  7  8
 9 10 11 12
13 14 15 16

{X}(f @ N) Y

⊢m←2 4⍴4/2 3
2 2 2 2
3 3 3 3      
      m×@2 4 ⊢ d
 1  2  3  4
10 12 14 16
 9 10 11 12
39 42 45 48

⊢d←4 4⍴⍳16
 1  2  3  4
 5  6  7  8
 9 10 11 12
13 14 15 16

{X}(f @ N) Y

⌽@1 3⊢'apple' 'berry' 'cherry'
 cherry  berry  apple
 
      ⌽¨@1 3⊢'apple' 'berry' 'cherry'
 elppa  berry  yrrehc

      2⌽@1 3 4⊢'apple' 'banana' 'berry' 'cherry' 
cherry  banana  apple  berry

{X}(f @ N) Y

g is a function that returns a Boolean array of shape ⍴Y

f is function to apply to elements of Y at indices ⍸g Y

(⌽¨@(2|≢¨))'dyalog' 'apl' 'is' 'great'
dyalog  lpa  is  taerg 

      (⌽@(2|≢¨))'dyalog' 'apl' 'is' 'great'
 dyalog  great  is  apl

{X}(f @ g) Y

Given an integer vector, double the first occurrence of the 
smallest number.

      (your solution)3 1 4 1 5 9
3 2 4 1 5 9
      

At Exercise

2×@(<\⊢=⌊/)⊢3 1 4 1 5 9
3 2 4 1 5 9

At Exercise Solution

X⍨

Constant

constant←⍨
      ⎕NC 'constant'
4 ⍝ operator

      seven←7⍨ 
      ⎕NC 'seven'
3 ⍝ function

Constant

'Hi'(1 2 3⍨)'Earth'
1 2 3

      (1 2 3⍨)'Hello'
1 2 3

Constant

42⍨¨'Life' 'Universe' 'Everything'
42 42 42
      (?9)⍨¨1 2∘.×1 2 3 4
7 7 7 7
7 7 7 7

Constant with other operators

normal code (⍺+⍵)*3

invalid train +*3

workaround 3*⍨+

workaround *∘3+

Constant in trains

normal code (⍺+⍵)*3

invalid train +*3

valid train +*3⍨

Constant in trains

normal code (⍺+⍵)*÷3

invalid train +*∘÷3

workaround (÷3)*⍨+

workaround 3*∘÷⍨+

Constant in trains

normal code (⍺+⍵)*÷3

invalid train +*∘÷3

valid train +*∘÷3⍨

valid train +*1÷3⍨

Constant in trains

mask←1 0 0 0 1 0 0 ⋄ data←'AbcdEfg'

'⎕'@{mask}data
⎕bcd⎕fg

mask{'⎕'@{⍺}⍵}data
VALUE ERROR

mask{'⎕'@{⍺}⍵}data
∧

Constant in operands

mask←1 0 0 0 1 0 0 ⋄ data←'AbcdEfg'

'⎕'@{mask}data
⎕bcd⎕fg

mask{'⎕'@(⍺⍨)⍵}data
⎕bcd⎕fg

Constant in operands

Given a matrix M, return 42 for every column.

      (your_solution) 3 4⍴⎕A
42 42 42 42

Constant Exercise

42⍨⌿3 4⍴⎕A
42 42 42 42

Constant Exercise Solution

f⌸

X⍸Y

⍸Y

≢Y

≠Y

A⍨

f@g  f@B  A@g  A@B

X⊣Y  X⊢Y  ⊢Y  ⊣Y

